Pensieve header: A unified verification notebook for the sl_2-portfolio project; continues pensieve://Projects/PPSA/nb/Verification.pdf.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
wdir = SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Portfolio"];
NotebookOpen[wdir <> "\\MakeSnips.nb"];
```

```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Initialization / Utilities

The “degree carrier / filtration parameter” is ℏ, and all “coupling constants” are proportional to it.

TD

```
$p = 8; $k = 2; (* $k can't be ∞ at least because of Faddeev-Quesne. *)
If[$k == 0, ϵ = 0, ϵ /: ϵ^k_ /; k > $k := 0]; (* $k=0 fails in Series[..{ϵ,...}] *)
SetAttributes[{SS, SST}, HoldAll];
TRule = {T_i_ → e^(ℏ t_i), T → e^(ℏ t)};
SS[ℰ_] := Block[{ℏ, ϵ}, (* Shielded Series *)
   Collect[Normal@Series[ℰ, {ℏ, 0, $p}], ℏ, Together] ];
SST[ℰ_] :=
  Block[{ℏ, ϵ}, Collect[Normal@Series[ℰ /. TRule, {ℏ, 0, $p}], ℏ, Together] ];
Simp[ℰ_, op_] := Collect[ℰ, _CU | _QU, op];
Simp[ℰ_] := Simp[ℰ, Collect[Normal@Series[#, {ℏ, 0, $p}], ℏ, Expand] &];
SimpT[ℰ_] :=
  Collect[ℰ, _CU | _QU, Collect[Normal@Series[# /. TRule, {ℏ, 0, $p}], ℏ, Expand] &];
```

Differential polynomials (DP):

Utils

```
DP_{α_→D_x_, β_→D_y_}[P_][λ_] :=
 Total[CoefficientRules[P, {α, β}] /. ({m_, n_} → c_) :→ c D[λ, {x, m}, {y, n}]]
```

DeclareAlgebra

QLImplementation

```
Unprotect[NonCommutativeMultiply]; Attributes[NonCommutativeMultiply] = {};
(NCM = NonCommutativeMultiply)[x_] := x;
NCM[x_, y_, z__] := (x ** y) ** z;
0 ** _ = _ ** 0 = 0;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, g, cp, CE, pow, k = 0,
    gs = Generators /. {opts}, cs = Centrals /. {opts} },
   (#_U = U@#) & /@ gs;
   gp = Alternatives @@ gs; gp = gp | gp_; (* gens *)
   sr = Flatten@Table[{g → ++k, g_i_ → {i, k}}, {g, gs}]; (* sorting → *)
   cp = Alternatives @@ cs; (* cents *)
   CE[ℰ_] := Collect[ℰ, _U, (Expand[#] /. ħ^d_ /; d > $p :> 0) &];
   U_i_[ℰ_] := ℰ /. {t : cp :> t_i, u_U :> Replace[u, x_ :> x_i, 1]};
   U_i_[NCM[]] = pow[ℰ_, 0] = U@{} = 1_U = U[];
   B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
   B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
   B[U@y_, U@x_] := CE[-B[U@x, U@y]];
   x_ ** 1_U := x; 1_U ** x_ := x;
   (a_. * x_U) ** (b_. * y_U) := If[a b === 0, 0, CE[a b (x ** y)]];
   U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
     U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
   U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := CE[c U@Table[l, {n}] ** U@{r}];
   U@{c_. * l : gp, r___} := CE[c U[l] ** U@{r}];
   U@{c_, r___} /; FreeQ[c, gp] := CE[c U@{r}];
   U@{l_Plus, r___} := CE[U@{#, r} & /@ l];
   U@{l_, r___} := U@{Expand[l], r};
   U[ℰ_NonCommutativeMultiply] := U /@ ℰ;
   𝕆_U[specs___, poly_] := Module[{sp, null, vs, us},
     sp = Replace[{specs}, l_List :> l_null, {1}];
     vs = Join @@ (First /@ sp);
     us = Join @@ (sp /. l_s_ :> (l /. x_i_ :> x_s));
     CE[Total[
        CoefficientRules[poly, vs] /. (p_ → c_) :> c U@(us^p)
       ]] /. x_null :> x];
   pow[ℰ_, n_] := pow[ℰ, n - 1] ** ℰ;
   𝕊_U[ℰ_, ss___Rule] := CE@Total[
     CoefficientRules[ℰ, First /@ {ss}] /.
      (p_ → c_) :> c NCM @@ MapThread[pow, {Last /@ {ss}, p}]];
   m_j_→k_[c_. * u_U] := CE[(c /. ((#_j → #_k) & /@ cs))
     DeleteCases[u, _(j|k)] ** U @@ Cases[u, w_j :> w_k] ** U @@ Cases[u, _k]];
   S_i_[c_. * u_U] := CE[(c /. S_i[U, Centrals])
     DeleteCases[u, _i] ** U_i[NCM @@ Reverse@Cases[u, x_i :> S@U@x]]] ]
```

DeclareMorphism

QLImplementation

```
DeclareMorphism[m_, U_ → V_, ongs_List, oncs_List: {}] := (
  Replace[ongs, (g_ → img_) ⧴ (m[U[g]] = img), {1}];
  m[1_U] = 1_V;
  m[U[g_i_]] := V_i[m[U@g]];
  m[U[vs__]] := NCM @@ (m /@ U /@ {vs});
  m[ℰ_] := Simp[ℰ /. oncs /. u_U ⧴ m[u]]; )
```

Meta-Operations

QLImplementation

```
m_{j_→j_} = Identity;
m_{j_→k_}[ℰ_Plus] := Simp[m_{j→k} /@ ℰ];
m_{is___,i_,j_→k_}[ℰ_] := m_{j→k}@m_{is,i→j}@ℰ;
S_i_[ℰ_Plus] := Simp[S_i /@ ℰ];
```

Implementing CU = $\mathcal{U}(sl_2^{\gamma\epsilon})$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[a_CU, y_CU] = -γ y_CU; B[x_CU, a_CU] = -γ x_CU;
B[x_CU, y_CU] = 2 ϵ a_CU - t 1_CU;
(S@CU@y = -y_CU; S@a_CU = -a_CU; S@x_CU = -x_CU;)
S_i_[CU, Centrals] = {t_i → -t_i};
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp // HL,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp // HL
  }] // Timing
```

```
{0.765625,
 {(28 t^2 γ^4 + 116 t γ^5 ϵ) CU[y, y, y, x, x] + <<21>> + CU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0}}
```

Verifying that S is an anti-homomorphism on CU:

```
With[{bas = CU /@ {y₁, a₁, x₁}},
 Table[HL@Simp[S₁[z1 ** z2] - S₁[z2] ** S₁[z1]],
  {z1, bas}, {z2, bas} ] ]
```

```
{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}
```

Verifying the involutivity of S on products of triples:

```
With[{bas = CU /@ {y₁, a₁, x₁}},
 Table[HL@Simp[z1 ** z2 ** z3 - S₁@S₁[z1 ** z2 ** z3]],
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying relabeling:

```
t₁ t₃ CU[y₁, a₁, x₂] + t₁ t₁ CU[y₁, a₂, x₂] // m₁→₃
```

$CU[a_2, x_2, y_3]\, t_3^2 + CU[x_2, y_3, a_3]\, t_3^2$

Verifying meta-associativity:

```
Module[{z, u},
 Table[u = CU[z⟦1⟧₁, z⟦2⟧₂, z⟦3⟧₃]; z → HL[m₁,₃→₃@m₂,₃→₃@u == m₂,₃→₃@m₁,₂→₂@u],
  {z, Tuples[{y, a, x}, 3]}]]
```

```
{{y, y, y} → True, {y, y, a} → True, {y, y, x} → True, {y, a, y} → True,
 {y, a, a} → True, {y, a, x} → True, {y, x, y} → True, {y, x, a} → True,
 {y, x, x} → True, {a, y, y} → True, {a, y, a} → True, {a, y, x} → True, {a, a, y} → True,
 {a, a, a} → True, {a, a, x} → True, {a, x, y} → True, {a, x, a} → True, {a, x, x} → True,
 {x, y, y} → True, {x, y, a} → True, {x, y, x} → True, {x, a, y} → True, {x, a, a} → True,
 {x, a, x} → True, {x, x, y} → True, {x, x, a} → True, {x, x, x} → True}
```

Implementing $QU = \mathcal{U}_q(sl_2^{\gamma\epsilon})$

Aside

```
Series[(1 - T ⅇ^(-2 e a h)) / h, {a, 0, 3}]
```

Aside

$$\frac{1-T}{h} + 2\,e\,T\,a - 2\left(e^2\,h\,T\right)a^2 + \frac{4}{3}\,e^3\,h^2\,T\,a^3 + O[a]^4$$

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q = SS[ⅇ^(γ ϵ ℏ)];
B[a_QU, y_QU] = -γ y_QU; B[x_QU, a_QU] = -γ QU@x;
B[x_QU, y_QU] = (q - 1) QU@{y, x} + 𝕆_QU[{a}, SS[(1 - T ⅇ^(-2 ϵ a ℏ)) / ℏ]];
(S@y_QU = 𝕆_QU[{a, y}, SS[-T^-1 ⅇ^(ℏ ϵ a) y]]; S@a_QU = -a_QU; S@x_QU = 𝕆_QU[{a, x}, SS[-ⅇ^(ℏ ϵ a) x]];)
S_i_[QU, Centrals] = {t_i → -t_i, T_i → T_i^-1};
```

```
With[{bas = QU /@ {y, a, x}}, Table[{z1, z2} → Simp[z1 ** z2 - z2 ** z1], {z1, bas}, {z2, bas}] ]
```

$$\{\{\{QU[y], QU[y]\} \to 0, \{QU[y], QU[a]\} \to \gamma\, QU[y], \{QU[y], QU[x]\} \to \frac{(-1+T)\, QU[]}{\hbar} - 2\, T\, \epsilon\, QU[a] - \gamma\, \epsilon\, \hbar\, QU[y, x]\},$$
$$\{\{QU[a], QU[y]\} \to -\gamma\, QU[y], \{QU[a], QU[a]\} \to 0, \{QU[a], QU[x]\} \to \gamma\, QU[x]\},$$
$$\{\{QU[x], QU[y]\} \to \frac{(1-T)\, QU[]}{\hbar} + 2\, T\, \epsilon\, QU[a] + \gamma\, \epsilon\, \hbar\, QU[y, x], \{QU[x], QU[a]\} \to -\gamma\, QU[x], \{QU[x], QU[x]\} \to 0\}\}$$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[HL[z1 ** (z2 ** z3) - (z1 ** z2) ** z3] // Simp,
  {z1, bas}, {z2, bas}, {z3, bas} ] ]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs @ $p=5, $k=2):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   HL[z1 ** (z2 ** z3) - rhs // Simp ]
  }] // Timing
```

$$\{6.3125, \{\left(\frac{28\,\gamma^4 - 56\, T\,\gamma^4 + 28\, T^2\,\gamma^4}{\hbar^2} + \frac{82\,\gamma^5\,\epsilon - 280 \ll 3\gg + 198\, T^2\,\gamma^5\,\epsilon}{\hbar}\right) QU[y, y, y, x, x] + \ll 18\gg + (1 + 8\,\gamma\,\epsilon\,\hbar)\, QU[\ll 1\gg], 0\}\}$$

Verifying that *S* is an anti-homomorphism on QU:

```
With[{bas = QU /@ {y₁, a₁, x₁}},
 Table[{z1, z2} → HL@Simp[S₁[z1 ** z2] - S₁[z2] ** S₁[z1]],
  {z1, bas}, {z2, bas} ] ]
```

$$\{\{\{QU[y_1], QU[y_1]\} \to 0, \{QU[y_1], QU[a_1]\} \to 0, \{QU[y_1], QU[x_1]\} \to 0\},$$
$$\{\{QU[a_1], QU[y_1]\} \to 0, \{QU[a_1], QU[a_1]\} \to 0, \{QU[a_1], QU[x_1]\} \to 0\},$$
$$\{\{QU[x_1], QU[y_1]\} \to 0, \{QU[x_1], QU[a_1]\} \to 0, \{QU[x_1], QU[x_1]\} \to 0\}\}$$

Verifying that $\text{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs @ $p=5, $k=2):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. TRule ⋃ {QU → CU}, ℏ → 0] - lhs] // HL
  }] // Timing
```

$$\{10.5313, \{48\, t\,\gamma^5\,\epsilon\, CU[y, y, y, x, x] + \ll 77\gg + CU[y, y, y, y, y, a, a, a, a, x, x, x, x],$$
$$2\left(\frac{4\,\gamma^5\,\epsilon}{\hbar} - \frac{8\, T\,\gamma^5\,\epsilon}{\hbar} + \frac{4\, T^2\,\gamma^5\,\epsilon}{\hbar}\right) QU[y, y, y, x, x] + \ll 217\gg + 8\,\gamma\,\epsilon\,\hbar\, QU[y, y, y, y, y, a, a, a, a, x, x, x, x], 0\}\}$$

Implementing θ

theta

```
DeclareMorphism[Cθ, CU → CU, {y → -x_CU, a → -a_CU, x → -y_CU}, {t → -t, T → T^-1}];
DeclareMorphism[Qθ, QU → QU, {y → 𝕆_QU[{a, x}, SS[-T^(-1/2) ⅇ^(ℏ ϵ a) x]],
  a → -a_QU, x → 𝕆_QU[{a, y}, SS[-T^(-1/2) ⅇ^(ℏ ϵ a) y]]}, {t → -t, T → T^-1}]
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[z → Cθ[z] → HL[Cθ[Cθ[z]]], {z, bas}] ]
```

{CU[y] → -CU[x] → CU[y], CU[a] → -CU[a] → CU[a], CU[x] → -CU[y] → CU[x]}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[Cθ[z1 ** z2] - Cθ[z1] ** Cθ[z2] // HL, {z1, bas}, {z2, bas}] ]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → Qθ[z] → HL[Simp[Qθ[Qθ[z]], PowerExpand]], {z, bas}] ]
```

$\left\{QU[y] \to -\frac{QU[x]}{\sqrt{T}} - \frac{\epsilon\hbar\, QU[a,x]}{\sqrt{T}} \to QU[y],\ QU[a] \to -QU[a] \to QU[a],\right.$

$\left. QU[x] \to \left(-\frac{1}{\sqrt{T}} + \frac{\gamma\epsilon\hbar}{\sqrt{T}}\right) QU[y] - \frac{\epsilon\hbar\, QU[y,a]}{\sqrt{T}} \to QU[x]\right\}$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[Simp[Qθ[z1 ** z2] - Qθ[z1] ** Qθ[z2]]], {z1, bas}, {z2, bas}] ]
```

{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}

The Asymmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

ADeq

$$\text{A}\mathbb{D}\$f = \gamma\left(\left(\text{Cosh}\left[\hbar\left(a\,e + \frac{\gamma\,e}{2} - \frac{t}{2}\right)\right] - \text{Cosh}\left[\hbar\sqrt{\left(\frac{t-\gamma\,e}{2}\right)^2 + e\,\omega}\right]\right)\Big/\left(\hbar\,e^{\hbar\,((a+\gamma)\,e-t/2)}\,\text{Sinh}\left[\frac{\gamma\,e\,\hbar}{2}\right]\left(a^2\,e + a\,\gamma\,e - a\,t - \omega\right)\right)\right);$$

Scaling behaviour of A𝔻$f:

```
HL@Simplify[A𝔻$f == ((A𝔻$f /. γ → 1) /. {e → γ e, a → γ^-1 a, ω → γ^-1 ω})]
```

```
True
```

```
HL@FullSimplify[
  A𝔻$f == ((A𝔻$f /. γ → 1) /. {ℏ → γ^2 ℏ, e → e / γ, a → a / γ, t → γ^-2 t, ω → γ^-3 ω})]
```

```
True
```

ADeq

```
A𝔻$ω = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a];
```

ADeq

```
DeclareMorphism[A𝔻, QU → CU,
  {a → a_CU, x → CU@x, y → 𝕊_CU[SS[A𝔻$f] /. e → ϵ, a → a_CU, ω → A𝔻$ω] ** y_CU}]
```

Verifying that the asymmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL[SimpT[A𝔻[z1 ** z2] - A𝔻[z1] ** A𝔻[z2]]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The Symmetric Dequantizator

Following pensieve://People/VanDerVeen/Dequant1.pdf.

SDeq

$$S\mathbb{D}\$g = \sqrt{\left(\left(2\gamma\left(\text{Cosh}\left[\frac{\hbar}{2}\sqrt{t^2+\gamma^2 e^2+4 e \varpi}\right]-\text{Cosh}\left[\frac{t-e\gamma-2 e a}{2/\hbar}\right]\right)\right)\Big/\left(\text{Sinh}\left[\frac{\gamma e \hbar}{2}\right]\left(t\,(2a+\gamma)-2a\,(a+\gamma)\,e+2\varpi\right)\hbar\right)\right)};$$

Verify agreement with the formulas in pensieve://People/VanDerVeen/Dequant1.pdf:

```
{SD$P = (Cosh[ħ ((e-t)/2 + e a)] - Cosh[ħ Sqrt[(t^2+e^2)/4 + e ϖ]]) / (ħ Sinh[-e ħ/2] (ϖ - e a^2 + (t - e) a + t/2)),
 Simplify[SD$P == (SD$P /. {a → -a - 1, t → -t})] // HL,
 PowerExpand@Simplify[(SD$P /. {ħ → γ^2 ħ, e → e/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}) ==
     SD$g (SD$g /. {a → -a - γ, t → -t})] // HL,
 SD$Q = Simplify[SD$P /. {a → c - 1/2}],
 Simplify[SD$Q == (SD$Q /. {c → -c, t → -t})] // HL,
 FullSimplify[SD$g == FullSimplify[
     Sqrt[SD$Q] /. c → a + 1/2 /. {ħ → γ^2 ħ, e → e/γ, a → a/γ, t → γ^-2 t, ϖ → γ^-3 ϖ}]] // HL
}
```

$$\left\{-\left(\left(\left(\mathrm{Cosh}\left[\left(a\,e+\frac{e-t}{2}\right)\hbar\right]-\mathrm{Cosh}\left[\sqrt{\frac{1}{4}\left(e^2+t^2\right)+e\,\varpi}\;\hbar\right]\right)\mathrm{Csch}\left[\frac{e\,\hbar}{2}\right]\right)\Big/\left(\left(-a^2 e+\frac{t}{2}+a\,(-e+t)+\varpi\right)\hbar\right)\right),\ \text{True},\ \text{True},\right.$$

$$\left.\left(4\left(-\mathrm{Cosh}\left[\frac{1}{2}\sqrt{e^2+t^2+4\,e\,\varpi}\;\hbar\right]+\mathrm{Cosh}\left[c\,e\,\hbar-\frac{t\,\hbar}{2}\right]\right)\mathrm{Csch}\left[\frac{e\,\hbar}{2}\right]\right)\Big/\left(\left(\left(-1+4c^2\right)e-4\,(c\,t+\varpi)\right)\hbar\right),\ \text{True},\ \text{True}\right\}$$

SDeq

```
SD$f = Simplify[ⅇ^(ħ (t/2-e a)) (SD$g /. {a → -a, t → -t})];
```

SDeq

```
SD$ϖ = γ CU[y, x] + ϵ CU[a, a] - (t - γ ϵ) CU[a] - t γ 1_CU / 2;
```

SDeq

```
DeclareMorphism[SD, QU → CU, {a → a_CU,
   x → S_CU[SS[SD$f] /. e → ϵ, a → a_CU, ϖ → SD$ϖ] ** x_CU,
   y → S_CU[SS[SD$g] /. e → ϵ, a → a_CU, ϖ → SD$ϖ] ** y_CU }]
```

Verifying the θ-symmetry:

```
Table[HL@SimpT[Cθ[SD[z]] == SD[Qθ[z]]], {z, QU /@ {y, a, x}}]
```

```
{True, True, True}
```

Verifying that the symmetric dequantizator is a homomorphism:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → HL@SimpT[SD[z1 ** z2] - SD[z1] ** SD[z2]], {z1, bas}, {z2, bas}] ]
```

```
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```

The representation ρ

rho

```
ρ@y_CU = ρ@y_QU = (0 0; ϵ 0); ρ@a_CU = ρ@a_QU = (γ 0; 0 0);
ρ@x_CU = (0 γ; 0 0); ρ@x_QU = SS@(0 (1 - ⅇ^(-γ ϵ ℏ)) / (ϵ ℏ); 0 0);
ρ[ⅇ^ℰ_] := MatrixExp[ρ[ℰ]];
ρ[ℰ_] :=
  (ℰ /. {t → γ ϵ, T → ⅇ^(ℏ γ ϵ)} /. (U : CU | QU)[u___] ⧴ Fold[Dot, (1 0; 0 1), ρ /@ U /@ {u}])
```

Verifying that ρ represents CU and QU:

```
Table[ρ[z1 ** z2] == ρ[z1].ρ[z2] // SS // HL,
 {U, {CU, QU}}, {z1, U /@ {y, a, x}}, {z2, U /@ {y, a, x}} ]
```

```
{{{True, True, True}, {True, True, True}, {True, True, True}},
 {{True, True, True}, {True, True, True}, {True, True, True}}}
```

Œ and the logoi Λ

Logoi from Pensieve://Talks/Toulouse-1705/DogmaDemo.nb and from Pensieve://Talks/Sydney-1708/ExtraDetails@@.nb.

CdsO

```
CU@Œ_CU[specs___, Q_, P_] := 𝕆_CU[specs, SS[ⅇ^Q P]];
QU@Œ_QU[specs___, Q_, P_] := 𝕆_QU[specs, SS[ⅇ^Q P]];
```

```
HL[ρ[ⅇ^(ξ CU@x)].ρ[ⅇ^(α CU@a)] == ρ[ⅇ^(α CU@a)].ρ[ⅇ^(ⅇ^(-γ α) ξ CU@x)]]
```

```
True
```

Logos

```
Λ_U_[{ξ_, α_}, {x, a}] := Œ_U[{a, x}, α a + ⅇ^(-γ α) ξ x, 1];
Λ_U_[{α_, η_}, {a, y}] := Œ_U[{y, a}, α a + ⅇ^(-γ α) η y, 1];
```

```
{Λ_#[{ξ, α}, {x, a}], lhs = #@Œ_#[{x, a}, ℏ (ξ x + α a), 1],
   HL[lhs == #@Λ_#[ℏ {ξ, α}, {x, a}]]} & /@ {CU, QU}
```

$$\Big\{\Big\{Œ_{CU}\big[\{a, x\}, a\,\alpha + e^{-\alpha\gamma} x\,\xi, 1\big],\ CU[\,] + \alpha\hbar\,CU[a] + \left(\xi\hbar - \alpha\gamma\xi\hbar^2\right) CU[x] + \frac{1}{2}\alpha^2\hbar^2\,CU[a, a] + \alpha\xi\hbar^2\,CU[a, x] + \frac{1}{2}\xi^2\hbar^2\,CU[x, x],\ \text{True}\Big\},\ \Big\{Œ_{QU}\big[\{a, x\}, a\,\alpha + e^{-\alpha\gamma} x\,\xi, 1\big],\ QU[\,] + \alpha\hbar\,QU[a] + \left(\xi\hbar - \alpha\gamma\xi\hbar^2\right) QU[x] + \frac{1}{2}\alpha^2\hbar^2\,QU[a, a] + \alpha\xi\hbar^2\,QU[a, x] + \frac{1}{2}\xi^2\hbar^2\,QU[x, x],\ \text{True}\Big\}\Big\}$$

```
{Λ#[{α, η}, {a, y}], lhs = #@Œ#[{a, y}, ℏ (η y + α a), 1],
   HL[lhs == #@Λ#[ℏ {α, η}, {a, y}]]} & /@ {CU, QU}
```

$\{\{\text{Œ}_{CU}[\{y, a\}, a\,\alpha + e^{-\alpha\gamma}\,y\,\eta, 1],$
$CU[\,] + \alpha\hbar\,CU[a] + (\eta\hbar - \alpha\gamma\eta\hbar^2)\,CU[y] + \frac{1}{2}\alpha^2\hbar^2\,CU[a, a] + \alpha\eta\hbar^2\,CU[y, a] + \frac{1}{2}\eta^2\hbar^2\,CU[y, y],$
$\text{True}\}, \{\text{Œ}_{QU}[\{y, a\}, a\,\alpha + e^{-\alpha\gamma}\,y\,\eta, 1], QU[\,] + \alpha\hbar\,QU[a] + (\eta\hbar - \alpha\gamma\eta\hbar^2)\,QU[y] +$
$\frac{1}{2}\alpha^2\hbar^2\,QU[a, a] + \alpha\eta\hbar^2\,QU[y, a] + \frac{1}{2}\eta^2\hbar^2\,QU[y, y], \text{True}\}\}$

Goal. In either U, compute $F = e^{-\eta y} e^{\xi x} e^{\eta y} e^{-\xi x}$. First compute $G = e^{\xi x} y e^{-\xi x}$, a finite sum. Now F satisfies the ODE $\partial_\eta F = \partial_\eta (e^{-\eta y} e^{\eta G}) = -yF + FG$ with initial conditions $F(\eta = 0) = 1$. So we set it up and solve:

```
With[{U = CU},
 Module[{G, F, fs, bs, e, b, es, sol},
  G = Echo@Simp[Table[ξ^k / k!, {k, 0, $k + 1}].NestList[Simp[B[x_U, #]] &, y_U, $k + 1]];
  fs = Echo@Flatten@Table[f_{l,i,j,k}[η], {l, 0, $k}, {i, 0, 1}, {j, 0, 1}, {k, 0, 1}];
  F = Echo[fs.(bs = fs /. f_{l_,i_,j_,k_}[η] :> ϵ^l U@{y^i, a^j, x^k})];
  es = Flatten[
    Table[Coefficient[e, b] == 0, {e, {F - 1_U /. η → 0, F ** G - y_U ** F - ∂_η F}}, {b, bs}]];
  sol = Echo@First[F /. DSolve[es, fs, η]];
  Echo[sol /. {e- → 1, U → Times}];
  Collect[sol /. {e- → 1, U → Times}, ϵ, Simplify]
 ]]
```

$-t\,\xi\,CU[\,] + 2\,\epsilon\,\xi\,CU[a] - \gamma\,\epsilon\,\xi^2\,CU[x] + CU[y]$

$\{f_{0,0,0,0}[\eta], f_{1,0,0,0}[\eta], f_{1,0,0,1}[\eta], f_{1,0,1,0}[\eta],$
$f_{1,0,1,1}[\eta], f_{1,1,0,0}[\eta], f_{1,1,0,1}[\eta], f_{1,1,1,0}[\eta], f_{1,1,1,1}[\eta]\}$

$CU[\,]\,f_{0,0,0,0}[\eta] + \epsilon\,CU[\,]\,f_{1,0,0,0}[\eta] + \epsilon\,CU[x]\,f_{1,0,0,1}[\eta] + \epsilon\,CU[a]\,f_{1,0,1,0}[\eta] + \epsilon\,CU[a, x]\,f_{1,0,1,1}[\eta] +$
$\epsilon\,CU[y]\,f_{1,1,0,0}[\eta] + \epsilon\,CU[y, x]\,f_{1,1,0,1}[\eta] + \epsilon\,CU[y, a]\,f_{1,1,1,0}[\eta] + \epsilon\,CU[y, a, x]\,f_{1,1,1,1}[\eta]$

$e^{-t\eta\xi}\,CU[\,] + \frac{1}{2}e^{-t\eta\xi}\,t\,\gamma\,\epsilon\,\eta^2\,\xi^2\,CU[\,] + 2\,e^{-t\eta\xi}\,\epsilon\,\eta\,\xi\,CU[a] - e^{-t\eta\xi}\,\gamma\,\epsilon\,\eta\,\xi^2\,CU[x] - e^{-t\eta\xi}\,\gamma\,\epsilon\,\eta^2\,\xi\,CU[y]$

$1 + 2\,a\,\epsilon\,\eta\,\xi - y\,\gamma\,\epsilon\,\eta^2\,\xi - x\,\gamma\,\epsilon\,\eta\,\xi^2 + \frac{1}{2}t\,\gamma\,\epsilon\,\eta^2\,\xi^2$

$1 + \frac{1}{2}\epsilon\,\eta\,\xi\,(4\,a + \gamma\,(-2\,y\,\eta - 2\,x\,\xi + t\,\eta\,\xi))$

Logos

```
Λ_U_[{ξ1_, η1_}, {x, y}] := Λ_U[{ξ1, η1}, {x, y}] = Module[{ξ, η, G, F, fs, f, bs, e, b, es},
   G = Simp[Table[ξ^k / k!, {k, 0, $k + 1}].NestList[Simp[B[x_U, #]] &, y_U, $k + 1]];
   fs = Flatten@Table[f_{l,i,j,k}[η], {l, 0, $k}, {i, 0, 1}, {j, 0, 1}, {k, 0, 1}];
   F = fs.(bs = fs /. f_{l_,i_,j_,k_}[η] :> ϵ^l U@{y^i, a^j, x^k});
   es = Flatten[
     Table[Coefficient[e, b] == 0, {e, {F - 1_U /. η → 0, F ** G - y_U ** F - ∂_η F}}, {b, bs}]];
   F = F /. DSolve[es, fs, η][[1]];
   Œ_U[{y, a, x},
     ξ x + η y + (U /. {CU → -t η ξ, QU → η ξ (1 - T) / ℏ}),
     F /. {e- → 1, U → Times}
    ] /. {ξ → ξ1, η → η1}];
```

```
{ΛCU[{ξ, η}, {x, y}], lhs = CU@ℭCU[{x, y}, ℏ (ξ x + η y), 1],
 HL[lhs == CU@ΛCU[ℏ {ξ, η}, {x, y}]]}
```

$\left\{\mathfrak{E}_{CU}\left[\{y, a, x\}, y\eta + x\xi - t\eta\xi, 1 + 2 a \in \eta\xi - y\gamma\in\eta^2\xi - x\gamma\in\eta\xi^2 + \frac{1}{2} t\gamma\in\eta^2\xi^2\right], \left(1 - t\eta\xi\hbar^2\right) CU[\,] + 2\in\eta\xi\hbar^2 CU[a] + \xi\hbar CU[x] + \eta\hbar CU[y] + \frac{1}{2}\xi^2\hbar^2 CU[x, x] + \eta\xi\hbar^2 CU[y, x] + \frac{1}{2}\eta^2\hbar^2 CU[y, y], \text{True}\right\}$

```
{ΛQU[{ξ, η}, {x, y}], lhs = QU@ℭQU[{x, y}, ℏ (ξ x + η y), 1],
 HL@SimpT[lhs == QU@ΛQU[ℏ {ξ, η}, {x, y}]]}
```

$\left\{\mathfrak{E}_{QU}\left[\{y, a, x\}, y\eta + x\xi + \frac{(1-T)\eta\xi}{\hbar}, 1 + 2 a T\in\eta\xi - \frac{1}{2}(-1+3T) y\gamma\in\eta^2\xi - \frac{1}{2}(-1+3T) x\gamma\in\eta\xi^2 + \frac{(-1+T)(-1+3T)\gamma\in\eta^2\xi^2}{4\hbar} + x y\gamma\in\eta\xi\hbar\right], \left(1 + \eta\xi\hbar - T\eta\xi\hbar\right) QU[\,] + 2 T\in\eta\xi\hbar^2 QU[a] + \xi\hbar QU[x] + \eta\hbar QU[y] + \frac{1}{2}\xi^2\hbar^2 QU[x, x] + \eta\xi\hbar^2 QU[y, x] + \frac{1}{2}\eta^2\hbar^2 QU[y, y], \text{True}\right\}$

```
{tt = Last[ΛCU[{ξ, η}, {x, y}]], Normal@Series[Log[tt], {ϵ, 0, $k}]}
```

$\left\{1 + 2 a\in\eta\xi - y\gamma\in\eta^2\xi - x\gamma\in\eta\xi^2 + \frac{1}{2} t\gamma\in\eta^2\xi^2, \in\left(2 a\eta\xi - y\gamma\eta^2\xi - x\gamma\eta\xi^2 + \frac{1}{2} t\gamma\eta^2\xi^2\right)\right\}$

```
{tt = Last[ΛQU[{ξ, η}, {x, y}]], Normal@Series[Log[tt], {ϵ, 0, $k}]}
```

$\left\{1 + 2 a T\in\eta\xi - \frac{1}{2}(-1+3T) y\gamma\in\eta^2\xi - \frac{1}{2}(-1+3T) x\gamma\in\eta\xi^2 + \frac{(-1+T)(-1+3T)\gamma\in\eta^2\xi^2}{4\hbar} + x y\gamma\in\eta\xi\hbar, \frac{1}{4\hbar}\in\left(\gamma\eta^2\xi^2 - 4T\gamma\eta^2\xi^2 + 3T^2\gamma\eta^2\xi^2 + 8 a T\eta\xi\hbar + 2 y\gamma\eta^2\xi\hbar - 6 T y\gamma\eta^2\xi\hbar + 2 x\gamma\eta\xi^2\hbar - 6 T x\gamma\eta\xi^2\hbar + 4 x y\gamma\eta\xi\hbar^2\right)\right\}$

Logos

```
Simp[ℭU_[specs___, Q_, P_]] :=
  ℭU[specs, ExpandNumerator@Together[Q], Collect[P, ϵ, ExpandNumerator@*Together]];
ΛU_[{υ1_, ω1_, δ_}, {u_, w_}] := Simp@Module[{υ, ω, yax, q, p, Q, d},
   {yax, q, p} = List @@ ΛU[{υ, ω}, {u, w}];
   ℭU[yax, Q = (υ u + ω w + δ u w + d υ ω) / (1 - d δ),
     Expand[(1 - d δ)^-1 e^-Q DP_{υ→D_u, ω→D_w}[p][e^Q]] /. {d → ∂_{υ,ω} q} /. {υ → υ1, ω → ω1}]
```

**{Λ_{CU}[{ξ, η, δ}, {x, y}], lhs = CU@$\mathbb{E}_{CU}$[{x, y}, ℏ (ξ x + η y + δ x y), 1],
HL[lhs == CU@Λ_{CU}[ℏ {ξ, η, δ}, {x, y}]]}**

$$\Big\{\mathbb{E}_{CU}\Big[\{y, a, x\}, \frac{x\,y\,\delta + y\,\eta + x\,\xi - t\,\eta\,\xi}{1 + t\,\delta}, \frac{1}{1 + t\,\delta} + \frac{1}{2\,(1 + t\,\delta)^5}\,\epsilon\,\big(4\,a\,\delta + 12\,a\,t\,\delta^2 + 4\,a\,x\,y\,\delta^2 + 2\,t\,\gamma\,\delta^2 - 8\,x\,y\,\gamma\,\delta^2 + 12\,a\,t^2\,\delta^3 + 8\,a\,t\,x\,y\,\delta^3 + 4\,t^2\,\gamma\,\delta^3 - 12\,t\,x\,y\,\gamma\,\delta^3 - 4\,x^2\,y^2\,\gamma\,\delta^3 + 4\,a\,t^3\,\delta^4 + 4\,a\,t^2\,x\,y\,\delta^4 + 2\,t^3\,\gamma\,\delta^4 - 4\,t^2\,x\,y\,\gamma\,\delta^4 - 3\,t\,x^2\,y^2\,\gamma\,\delta^4 + 4\,a\,y\,\delta\,\eta - 4\,y\,\gamma\,\delta\,\eta + 8\,a\,t\,y\,\delta^2\,\eta - 4\,t\,y\,\gamma\,\delta^2\,\eta - 6\,x\,y^2\,\gamma\,\delta^2\,\eta + 4\,a\,t^2\,y\,\delta^3\,\eta - 4\,t\,x\,y^2\,\gamma\,\delta^3\,\eta - 2\,y^2\,\gamma\,\delta\,\eta^2 - t\,y^2\,\gamma\,\delta^2\,\eta^2 + 4\,a\,x\,\delta\,\xi - 4\,x\,\gamma\,\delta\,\xi + 8\,a\,t\,x\,\delta^2\,\xi - 4\,t\,x\,\gamma\,\delta^2\,\xi - 6\,x^2\,y\,\gamma\,\delta^2\,\xi + 4\,a\,t^2\,x\,\delta^3\,\xi - 4\,t\,x^2\,y\,\gamma\,\delta^3\,\xi + 4\,a\,\eta\,\xi + 8\,a\,t\,\delta\,\eta\,\xi + 4\,t\,\gamma\,\delta\,\eta\,\xi - 8\,x\,y\,\gamma\,\delta\,\eta\,\xi + 4\,a\,t^2\,\delta^2\,\eta\,\xi + 4\,t^2\,\gamma\,\delta^2\,\eta\,\xi - 4\,t\,x\,y\,\gamma\,\delta^2\,\eta\,\xi - 2\,y\,\gamma\,\eta^2\,\xi - 2\,x^2\,\gamma\,\delta\,\xi^2 - t\,x^2\,\gamma\,\delta^2\,\xi^2 - 2\,x\,\gamma\,\eta\,\xi^2 + t\,\gamma\,\eta^2\,\xi^2\big)\Big],$$

$$\left(1 - t\,\delta\,\hbar + t^2\,\delta^2\,\hbar^2 + t\,\gamma\,\delta^2\,\epsilon\,\hbar^2 - t\,\eta\,\xi\,\hbar^2\right) \mathrm{CU}[\,] + \left(2\,\delta\,\epsilon\,\hbar - 4\,t\,\delta^2\,\epsilon\,\hbar^2 + 2\,\epsilon\,\eta\,\xi\,\hbar^2\right) \mathrm{CU}[a] +$$
$$\left(\xi\,\hbar - 2\,t\,\delta\,\xi\,\hbar^2 - 2\,\gamma\,\delta\,\epsilon\,\xi\,\hbar^2\right) \mathrm{CU}[x] +$$
$$\left(\eta\,\hbar - 2\,t\,\delta\,\eta\,\hbar^2 - 2\,\gamma\,\delta\,\epsilon\,\eta\,\hbar^2\right) \mathrm{CU}[y] +$$
$$4\,\delta\,\epsilon\,\xi\,\hbar^2\,\mathrm{CU}[a, x] +$$
$$\frac{1}{2}\,\xi^2\,\hbar^2\,\mathrm{CU}[x, x] +$$
$$4\,\delta\,\epsilon\,\eta\,\hbar^2\,\mathrm{CU}[y, a] +$$
$$\left(\delta\,\hbar - 2\,t\,\delta^2\,\hbar^2 - 4\,\gamma\,\delta^2\,\epsilon\,\hbar^2 + \eta\,\xi\,\hbar^2\right) \mathrm{CU}[y, x] +$$
$$\frac{1}{2}\,\eta^2\,\hbar^2\,\mathrm{CU}[y, y] +$$
$$4\,\delta^2\,\epsilon\,\hbar^2\,\mathrm{CU}[y, a, x] +$$
$$\delta\,\xi\,\hbar^2\,\mathrm{CU}[y, x, x] +$$
$$\delta\,\eta\,\hbar^2\,\mathrm{CU}[y, y, x] +$$
$$\frac{1}{2}\,\delta^2\,\hbar^2\,\mathrm{CU}[y, y, x, x], \text{True}\Big\}$$

```
{ΛQU[{ξ, η, δ}, {x, y}], lhs = QU@𝔈QU[{x, y}, ℏ (ξ x + η y + δ x y), 1],
 HL@SimpT[lhs == QU@ΛQU[ℏ {ξ, η, δ}, {x, y}]]}
```

$$\Big\{\mathbb{E}_{QU}\Big[\{y, a, x\}, \frac{\eta\,\xi - T\,\eta\,\xi + x\,y\,\delta\,\hbar + y\,\eta\,\hbar + x\,\xi\,\hbar}{-\delta + T\,\delta + \hbar},$$

$$\frac{\hbar}{-\delta + T\,\delta + \hbar} + \frac{1}{4\,(-\delta + T\,\delta + \hbar)^5}\in\Big(-8\,a\,T\,\delta^4\,\hbar^2 + 24\,a\,T^2\,\delta^4\,\hbar^2 - 24\,a\,T^3\,\delta^4\,\hbar^2 + 8\,a\,T^4\,\delta^4\,\hbar^2 + 2\,\gamma\,\delta^4\,\hbar^2 -$$

$$12\,T\,\gamma\,\delta^4\,\hbar^2 + 24\,T^2\,\gamma\,\delta^4\,\hbar^2 - 20\,T^3\,\gamma\,\delta^4\,\hbar^2 + 6\,T^4\,\gamma\,\delta^4\,\hbar^2 + 24\,a\,T\,\delta^3\,\hbar^3 - 48\,a\,T^2\,\delta^3\,\hbar^3 +$$

$$24\,a\,T^3\,\delta^3\,\hbar^3 - 4\,\gamma\,\delta^3\,\hbar^3 + 20\,T\,\gamma\,\delta^3\,\hbar^3 - 28\,T^2\,\gamma\,\delta^3\,\hbar^3 + 12\,T^3\,\gamma\,\delta^3\,\hbar^3 + 8\,a\,T\,x\,y\,\delta^4\,\hbar^3 -$$

$$16\,a\,T^2\,x\,y\,\delta^4\,\hbar^3 + 8\,a\,T^3\,x\,y\,\delta^4\,\hbar^3 - 8\,T\,x\,y\,\gamma\,\delta^4\,\hbar^3 + 16\,T^2\,x\,y\,\gamma\,\delta^4\,\hbar^3 - 8\,T^3\,x\,y\,\gamma\,\delta^4\,\hbar^3 +$$

$$8\,a\,T\,y\,\delta^3\,\eta\,\hbar^3 - 16\,a\,T^2\,y\,\delta^3\,\eta\,\hbar^3 + 8\,a\,T^3\,y\,\delta^3\,\eta\,\hbar^3 + 8\,a\,T\,x\,\delta^3\,\xi\,\hbar^3 - 16\,a\,T^2\,x\,\delta^3\,\xi\,\hbar^3 +$$

$$8\,a\,T^3\,x\,\delta^3\,\xi\,\hbar^3 + 8\,a\,T\,\delta^2\,\eta\,\xi\,\hbar^3 - 16\,a\,T^2\,\delta^2\,\eta\,\xi\,\hbar^3 + 8\,a\,T^3\,\delta^2\,\eta\,\xi\,\hbar^3 - 4\,\gamma\,\delta^2\,\eta\,\xi\,\hbar^3 +$$

$$20\,T\,\gamma\,\delta^2\,\eta\,\xi\,\hbar^3 - 28\,T^2\,\gamma\,\delta^2\,\eta\,\xi\,\hbar^3 + 12\,T^3\,\gamma\,\delta^2\,\eta\,\xi\,\hbar^3 - 24\,a\,T\,\delta^2\,\hbar^4 + 24\,a\,T^2\,\delta^2\,\hbar^4 +$$

$$2\,\gamma\,\delta^2\,\hbar^4 - 8\,T\,\gamma\,\delta^2\,\hbar^4 + 6\,T^2\,\gamma\,\delta^2\,\hbar^4 - 16\,a\,T\,x\,y\,\delta^3\,\hbar^4 + 16\,a\,T^2\,x\,y\,\delta^3\,\hbar^4 + 24\,T\,x\,y\,\gamma\,\delta^3\,\hbar^4 -$$

$$24\,T^2\,x\,y\,\gamma\,\delta^3\,\hbar^4 + x^2\,y^2\,\gamma\,\delta^4\,\hbar^4 + 4\,T\,x^2\,y^2\,\gamma\,\delta^4\,\hbar^4 - 5\,T^2\,x^2\,y^2\,\gamma\,\delta^4\,\hbar^4 - 16\,a\,T\,y\,\delta^2\,\eta\,\hbar^4 +$$

$$16\,a\,T^2\,y\,\delta^2\,\eta\,\hbar^4 - 4\,y\,\gamma\,\delta^2\,\eta\,\hbar^4 + 16\,T\,y\,\gamma\,\delta^2\,\eta\,\hbar^4 - 12\,T^2\,y\,\gamma\,\delta^2\,\eta\,\hbar^4 + 8\,T\,x\,y^2\,\gamma\,\delta^3\,\eta\,\hbar^4 -$$

$$8\,T^2\,x\,y^2\,\gamma\,\delta^3\,\eta\,\hbar^4 - y^2\,\gamma\,\delta^2\,\eta^2\,\hbar^4 + 4\,T\,y^2\,\gamma\,\delta^2\,\eta^2\,\hbar^4 - 3\,T^2\,y^2\,\gamma\,\delta^2\,\eta^2\,\hbar^4 - 16\,a\,T\,x\,\delta^2\,\xi\,\hbar^4 +$$

$$16\,a\,T^2\,x\,\delta^2\,\xi\,\hbar^4 - 4\,x\,\gamma\,\delta^2\,\xi\,\hbar^4 + 16\,T\,x\,\gamma\,\delta^2\,\xi\,\hbar^4 - 12\,T^2\,x\,\gamma\,\delta^2\,\xi\,\hbar^4 + 8\,T\,x^2\,y\,\gamma\,\delta^3\,\xi\,\hbar^4 -$$

$$8\,T^2\,x^2\,y\,\gamma\,\delta^3\,\xi\,\hbar^4 - 16\,a\,T\,\delta\,\eta\,\xi\,\hbar^4 + 16\,a\,T^2\,\delta\,\eta\,\xi\,\hbar^4 + 4\,\gamma\,\delta\,\eta\,\xi\,\hbar^4 - 16\,T\,\gamma\,\delta\,\eta\,\xi\,\hbar^4 +$$

$$12\,T^2\,\gamma\,\delta\,\eta\,\xi\,\hbar^4 + 8\,T\,x\,y\,\gamma\,\delta^2\,\eta\,\xi\,\hbar^4 - 8\,T^2\,x\,y\,\gamma\,\delta^2\,\eta\,\xi\,\hbar^4 - x^2\,\gamma\,\delta^2\,\xi^2\,\hbar^4 + 4\,T\,x^2\,\gamma\,\delta^2\,\xi^2\,\hbar^4 -$$

$$3\,T^2\,x^2\,\gamma\,\delta^2\,\xi^2\,\hbar^4 + \gamma\,\eta^2\,\xi^2\,\hbar^4 - 4\,T\,\gamma\,\eta^2\,\xi^2\,\hbar^4 + 3\,T^2\,\gamma\,\eta^2\,\xi^2\,\hbar^4 + 8\,a\,T\,\delta\,\hbar^5 + 8\,a\,T\,x\,y\,\delta^2\,\hbar^5 -$$

$$4\,x\,y\,\gamma\,\delta^2\,\hbar^5 - 12\,T\,x\,y\,\gamma\,\delta^2\,\hbar^5 - 4\,x^2\,y^2\,\gamma\,\delta^3\,\hbar^5 - 4\,T\,x^2\,y^2\,\gamma\,\delta^3\,\hbar^5 + 8\,a\,T\,y\,\delta\,\eta\,\hbar^5 + 4\,y\,\gamma\,\delta\,\eta\,\hbar^5 -$$

$$12\,T\,y\,\gamma\,\delta\,\eta\,\hbar^5 - 2\,x\,y^2\,\gamma\,\delta^2\,\eta\,\hbar^5 - 10\,T\,x\,y^2\,\gamma\,\delta^2\,\eta\,\hbar^5 + 2\,y^2\,\gamma\,\delta\,\eta^2\,\hbar^5 - 6\,T\,y^2\,\gamma\,\delta\,\eta^2\,\hbar^5 +$$

$$8\,a\,T\,x\,\delta\,\xi\,\hbar^5 + 4\,x\,\gamma\,\delta\,\xi\,\hbar^5 - 12\,T\,x\,\gamma\,\delta\,\xi\,\hbar^5 - 2\,x^2\,y\,\gamma\,\delta^2\,\xi\,\hbar^5 - 10\,T\,x^2\,y\,\gamma\,\delta^2\,\xi\,\hbar^5 + 8\,a\,T\,\eta\,\xi\,\hbar^5 -$$

$$16\,T\,x\,y\,\gamma\,\delta\,\eta\,\xi\,\hbar^5 + 2\,y\,\gamma\,\eta^2\,\xi\,\hbar^5 - 6\,T\,y\,\gamma\,\eta^2\,\xi\,\hbar^5 + 2\,x^2\,\gamma\,\delta\,\xi^2\,\hbar^5 - 6\,T\,x^2\,\gamma\,\delta\,\xi^2\,\hbar^5 + 2\,x\,\gamma\,\eta\,\xi^2\,\hbar^5 -$$

$$6\,T\,x\,\gamma\,\eta\,\xi^2\,\hbar^5 + 4\,x\,y\,\gamma\,\delta\,\hbar^6 + 4\,x^2\,y^2\,\gamma\,\delta^2\,\hbar^6 + 4\,x\,y^2\,\gamma\,\delta\,\eta\,\hbar^6 + 4\,x^2\,y\,\gamma\,\delta\,\xi\,\hbar^6 + 4\,x\,y\,\gamma\,\eta\,\xi\,\hbar^6\Big)\Big],$$

$$\left(1 + \delta - T\,\delta + \delta^2 - 2\,T\,\delta^2 + T^2\,\delta^2 + \frac{1}{2}\,\gamma\,\delta^2\in\hbar - 2\,T\,\gamma\,\delta^2\in\hbar + \frac{3}{2}\,T^2\,\gamma\,\delta^2\in\hbar + \eta\,\xi\,\hbar - T\,\eta\,\xi\,\hbar\right)$$

$$QU[\,] + \left(2\,T\,\delta\in\hbar + 4\,T\,\delta^2\in\hbar - 4\,T^2\,\delta^2\in\hbar + 2\,T\in\eta\,\xi\,\hbar^2\right) QU[a] +$$

$$\left(\xi\,\hbar + 2\,\delta\,\xi\,\hbar - 2\,T\,\delta\,\xi\,\hbar + \gamma\,\delta\in\xi\,\hbar^2 - 3\,T\,\gamma\,\delta\in\xi\,\hbar^2\right) QU[x] +$$

$$\left(\eta\,\hbar + 2\,\delta\,\eta\,\hbar - 2\,T\,\delta\,\eta\,\hbar + \gamma\,\delta\in\eta\,\hbar^2 - 3\,T\,\gamma\,\delta\in\eta\,\hbar^2\right) QU[y] +$$

$$4\,T\,\delta\in\xi\,\hbar^2\,QU[a, x] + \frac{1}{2}\,\xi^2\,\hbar^2\,QU[x, x] +$$

$$4\,T\,\delta\in\eta\,\hbar^2\,QU[y, a] +$$

$$\left(\delta\,\hbar + 2\,\delta^2\,\hbar - 2\,T\,\delta^2\,\hbar + \gamma\,\delta\in\hbar^2 + 4\,\gamma\,\delta^2\in\hbar^2 - 8\,T\,\gamma\,\delta^2\in\hbar^2 + \eta\,\xi\,\hbar^2\right) QU[y, x] +$$

$$\frac{1}{2}\,\eta^2\,\hbar^2\,QU[y, y] + 4\,T\,\delta^2\in\hbar^2\,QU[y, a, x] +$$

$$\delta\,\xi\,\hbar^2\,QU[y, x, x] +$$

$$\delta\,\eta\,\hbar^2\,QU[y, y, x] +$$

$$\frac{1}{2}\,\delta^2\,\hbar^2\,QU[y, y, x, x], \text{True}\Big\}$$

```
{tt = Last[ΛCU[{ξ, η, δ}, {x, y}]], Normal@Series[Log[tt], {ϵ, 0, $k}]}
```

$$\left\{\frac{1}{1+t\,\delta}+\frac{1}{2\left(1+t\,\delta\right)^{5}}\epsilon\left(4\,a\,\delta+12\,a\,t\,\delta^{2}+4\,a\,x\,y\,\delta^{2}+2\,t\,\gamma\,\delta^{2}-8\,x\,y\,\gamma\,\delta^{2}+12\,a\,t^{2}\,\delta^{3}+8\,a\,t\,x\,y\,\delta^{3}+4\,t^{2}\,\gamma\,\delta^{3}-12\,t\,x\,y\,\gamma\,\delta^{3}-4\,x^{2}\,y^{2}\,\gamma\,\delta^{3}+4\,a\,t^{3}\,\delta^{4}+4\,a\,t^{2}\,x\,y\,\delta^{4}+2\,t^{3}\,\gamma\,\delta^{4}-4\,t^{2}\,x\,y\,\gamma\,\delta^{4}-3\,t\,x^{2}\,y^{2}\,\gamma\,\delta^{4}+4\,a\,y\,\delta\,\eta-4\,y\,\gamma\,\delta\,\eta+8\,a\,t\,y\,\delta^{2}\,\eta-4\,t\,y\,\gamma\,\delta^{2}\,\eta-6\,x\,y^{2}\,\gamma\,\delta^{2}\,\eta+4\,a\,t^{2}\,y\,\delta^{3}\,\eta-4\,t\,x\,y^{2}\,\gamma\,\delta^{3}\,\eta-2\,y^{2}\,\gamma\,\delta\,\eta^{2}-t\,y^{2}\,\gamma\,\delta^{2}\,\eta^{2}+4\,a\,x\,\delta\,\xi-4\,x\,\gamma\,\delta\,\xi+8\,a\,t\,x\,\delta^{2}\,\xi-4\,t\,x\,\gamma\,\delta^{2}\,\xi-6\,x^{2}\,y\,\gamma\,\delta^{2}\,\xi+4\,a\,t^{2}\,x\,\delta^{3}\,\xi-4\,t\,x^{2}\,y\,\gamma\,\delta^{3}\,\xi+4\,a\,\eta\,\xi+8\,a\,t\,\delta\,\eta\,\xi+4\,t\,\gamma\,\delta\,\eta\,\xi-8\,x\,y\,\gamma\,\delta\,\eta\,\xi+4\,a\,t^{2}\,\delta^{2}\,\eta\,\xi+4\,t^{2}\,\gamma\,\delta^{2}\,\eta\,\xi-4\,t\,x\,y\,\gamma\,\delta^{2}\,\eta\,\xi-2\,y\,\gamma\,\eta^{2}\,\xi-2\,x^{2}\,\gamma\,\delta\,\xi^{2}-t\,x^{2}\,\gamma\,\delta^{2}\,\xi^{2}-2\,x\,\gamma\,\eta\,\xi^{2}+t\,\gamma\,\eta^{2}\,\xi^{2}\right),\right.$$

$$\left.\frac{1}{2\left(1+t\,\delta\right)^{4}}\epsilon\left(4\,a\,\delta+12\,a\,t\,\delta^{2}+4\,a\,x\,y\,\delta^{2}+2\,t\,\gamma\,\delta^{2}-8\,x\,y\,\gamma\,\delta^{2}+12\,a\,t^{2}\,\delta^{3}+8\,a\,t\,x\,y\,\delta^{3}+4\,t^{2}\,\gamma\,\delta^{3}-12\,t\,x\,y\,\gamma\,\delta^{3}-4\,x^{2}\,y^{2}\,\gamma\,\delta^{3}+4\,a\,t^{3}\,\delta^{4}+4\,a\,t^{2}\,x\,y\,\delta^{4}+2\,t^{3}\,\gamma\,\delta^{4}-4\,t^{2}\,x\,y\,\gamma\,\delta^{4}-3\,t\,x^{2}\,y^{2}\,\gamma\,\delta^{4}+4\,a\,y\,\delta\,\eta-4\,y\,\gamma\,\delta\,\eta+8\,a\,t\,y\,\delta^{2}\,\eta-4\,t\,y\,\gamma\,\delta^{2}\,\eta-6\,x\,y^{2}\,\gamma\,\delta^{2}\,\eta+4\,a\,t^{2}\,y\,\delta^{3}\,\eta-4\,t\,x\,y^{2}\,\gamma\,\delta^{3}\,\eta-2\,y^{2}\,\gamma\,\delta\,\eta^{2}-t\,y^{2}\,\gamma\,\delta^{2}\,\eta^{2}+4\,a\,x\,\delta\,\xi-4\,x\,\gamma\,\delta\,\xi+8\,a\,t\,x\,\delta^{2}\,\xi-4\,t\,x\,\gamma\,\delta^{2}\,\xi-6\,x^{2}\,y\,\gamma\,\delta^{2}\,\xi+4\,a\,t^{2}\,x\,\delta^{3}\,\xi-4\,t\,x^{2}\,y\,\gamma\,\delta^{3}\,\xi+4\,a\,\eta\,\xi+8\,a\,t\,\delta\,\eta\,\xi+4\,t\,\gamma\,\delta\,\eta\,\xi-8\,x\,y\,\gamma\,\delta\,\eta\,\xi+4\,a\,t^{2}\,\delta^{2}\,\eta\,\xi+4\,t^{2}\,\gamma\,\delta^{2}\,\eta\,\xi-4\,t\,x\,y\,\gamma\,\delta^{2}\,\eta\,\xi-2\,y\,\gamma\,\eta^{2}\,\xi-2\,x^{2}\,\gamma\,\delta\,\xi^{2}-t\,x^{2}\,\gamma\,\delta^{2}\,\xi^{2}-2\,x\,\gamma\,\eta\,\xi^{2}+t\,\gamma\,\eta^{2}\,\xi^{2}\right)+\mathrm{Log}\left[\frac{1}{1+t\,\delta}\right]\right\}$$

```
{tt = Last[ΛQU[{ξ, η, δ}, {x, y}]], Normal@Series[Log[tt], {ϵ, 0, $k}]}
```

$$
\begin{aligned}
&\Big\{\frac{\hbar}{-\delta+T\delta+\hbar}+\\
&\frac{1}{4\,(-\delta+T\delta+\hbar)^5}\,\epsilon\,\big(-8\,a\,T\,\delta^4\hbar^2+24\,a\,T^2\delta^4\hbar^2-24\,a\,T^3\delta^4\hbar^2+8\,a\,T^4\delta^4\hbar^2+2\,\gamma\,\delta^4\hbar^2-12\,T\,\gamma\,\delta^4\hbar^2+\\
&\quad 24\,T^2\gamma\,\delta^4\hbar^2-20\,T^3\gamma\,\delta^4\hbar^2+6\,T^4\gamma\,\delta^4\hbar^2+24\,a\,T\,\delta^3\hbar^3-48\,a\,T^2\delta^3\hbar^3+24\,a\,T^3\delta^3\hbar^3-\\
&\quad 4\,\gamma\,\delta^3\hbar^3+20\,T\,\gamma\,\delta^3\hbar^3-28\,T^2\gamma\,\delta^3\hbar^3+12\,T^3\gamma\,\delta^3\hbar^3+8\,a\,T\,x\,y\,\delta^4\hbar^3-16\,a\,T^2x\,y\,\delta^4\hbar^3+\\
&\quad 8\,a\,T^3x\,y\,\delta^4\hbar^3-8\,T\,x\,y\,\gamma\,\delta^4\hbar^3+16\,T^2x\,y\,\gamma\,\delta^4\hbar^3-8\,T^3x\,y\,\gamma\,\delta^4\hbar^3+8\,a\,T\,y\,\delta^3\eta\,\hbar^3-\\
&\quad 16\,a\,T^2y\,\delta^3\eta\,\hbar^3+8\,a\,T^3y\,\delta^3\eta\,\hbar^3+8\,a\,T\,x\,\delta^3\xi\,\hbar^3-16\,a\,T^2x\,\delta^3\xi\,\hbar^3+8\,a\,T^3x\,\delta^3\xi\,\hbar^3+\\
&\quad 8\,a\,T\,\delta^2\eta\,\xi\,\hbar^3-16\,a\,T^2\delta^2\eta\,\xi\,\hbar^3+8\,a\,T^3\delta^2\eta\,\xi\,\hbar^3-4\,\gamma\,\delta^2\eta\,\xi\,\hbar^3+20\,T\,\gamma\,\delta^2\eta\,\xi\,\hbar^3-\\
&\quad 28\,T^2\gamma\,\delta^2\eta\,\xi\,\hbar^3+12\,T^3\gamma\,\delta^2\eta\,\xi\,\hbar^3-24\,a\,T\,\delta^2\hbar^4+24\,a\,T^2\delta^2\hbar^4+2\,\gamma\,\delta^2\hbar^4-8\,T\,\gamma\,\delta^2\hbar^4+\\
&\quad 6\,T^2\gamma\,\delta^2\hbar^4-16\,a\,T\,x\,y\,\delta^3\hbar^4+16\,a\,T^2x\,y\,\delta^3\hbar^4+24\,T\,x\,y\,\gamma\,\delta^3\hbar^4-24\,T^2x\,y\,\gamma\,\delta^3\hbar^4+\\
&\quad x^2y^2\gamma\,\delta^4\hbar^4+4\,T\,x^2y^2\gamma\,\delta^4\hbar^4-5\,T^2x^2y^2\gamma\,\delta^4\hbar^4-16\,a\,T\,y\,\delta^2\eta\,\hbar^4+16\,a\,T^2y\,\delta^2\eta\,\hbar^4-\\
&\quad 4\,y\,\gamma\,\delta^2\eta\,\hbar^4+16\,T\,y\,\gamma\,\delta^2\eta\,\hbar^4-12\,T^2y\,\gamma\,\delta^2\eta\,\hbar^4+8\,T\,x\,y^2\gamma\,\delta^3\eta\,\hbar^4-8\,T^2x\,y^2\gamma\,\delta^3\eta\,\hbar^4-\\
&\quad y^2\gamma\,\delta^2\eta^2\hbar^4+4\,T\,y^2\gamma\,\delta^2\eta^2\hbar^4-3\,T^2y^2\gamma\,\delta^2\eta^2\hbar^4-16\,a\,T\,x\,\delta^2\xi\,\hbar^4+16\,a\,T^2x\,\delta^2\xi\,\hbar^4-\\
&\quad 4\,x\,\gamma\,\delta^2\xi\,\hbar^4+16\,T\,x\,\gamma\,\delta^2\xi\,\hbar^4-12\,T^2x\,\gamma\,\delta^2\xi\,\hbar^4+8\,T\,x^2y\,\gamma\,\delta^3\xi\,\hbar^4-8\,T^2x^2y\,\gamma\,\delta^3\xi\,\hbar^4-\\
&\quad 16\,a\,T\,\delta\,\eta\,\xi\,\hbar^4+16\,a\,T^2\delta\,\eta\,\xi\,\hbar^4+4\,\gamma\,\delta\,\eta\,\xi\,\hbar^4-16\,T\,\gamma\,\delta\,\eta\,\xi\,\hbar^4+12\,T^2\gamma\,\delta\,\eta\,\xi\,\hbar^4+\\
&\quad 8\,T\,x\,y\,\gamma\,\delta^2\eta\,\xi\,\hbar^4-8\,T^2x\,y\,\gamma\,\delta^2\eta\,\xi\,\hbar^4-x^2\gamma\,\delta^2\xi^2\hbar^4+4\,T\,x^2\gamma\,\delta^2\xi^2\hbar^4-3\,T^2x^2\gamma\,\delta^2\xi^2\hbar^4+\\
&\quad \gamma\,\eta^2\xi^2\hbar^4-4\,T\,\gamma\,\eta^2\xi^2\hbar^4+3\,T^2\gamma\,\eta^2\xi^2\hbar^4+8\,a\,T\,\delta\,\hbar^5+8\,a\,T\,x\,y\,\delta^2\hbar^5-4\,x\,y\,\gamma\,\delta^2\hbar^5-\\
&\quad 12\,T\,x\,y\,\gamma\,\delta^2\hbar^5-4\,x^2y^2\gamma\,\delta^3\hbar^5-4\,T\,x^2y^2\gamma\,\delta^3\hbar^5+8\,a\,T\,y\,\delta\,\eta\,\hbar^5+4\,y\,\gamma\,\delta\,\eta\,\hbar^5-\\
&\quad 12\,T\,y\,\gamma\,\delta\,\eta\,\hbar^5-2\,x\,y^2\gamma\,\delta^2\eta\,\hbar^5-10\,T\,x\,y^2\gamma\,\delta^2\eta\,\hbar^5+2\,y^2\gamma\,\delta\,\eta^2\hbar^5-6\,T\,y^2\gamma\,\delta\,\eta^2\hbar^5+\\
&\quad 8\,a\,T\,x\,\delta\,\xi\,\hbar^5+4\,x\,\gamma\,\delta\,\xi\,\hbar^5-12\,T\,x\,\gamma\,\delta\,\xi\,\hbar^5-2\,x^2y\,\gamma\,\delta^2\xi\,\hbar^5-10\,T\,x^2y\,\gamma\,\delta^2\xi\,\hbar^5+8\,a\,T\,\eta\,\xi\,\hbar^5-\\
&\quad 16\,T\,x\,y\,\gamma\,\delta\,\eta\,\xi\,\hbar^5+2\,y\,\gamma\,\eta^2\xi\,\hbar^5-6\,T\,y\,\gamma\,\eta^2\xi\,\hbar^5+2\,x^2\gamma\,\delta\,\xi^2\hbar^5-6\,T\,x^2\gamma\,\delta\,\xi^2\hbar^5+2\,x\,\gamma\,\eta\,\xi^2\hbar^5-\\
&\quad 6\,T\,x\,\gamma\,\eta\,\xi^2\hbar^5+4\,x\,y\,\gamma\,\delta\,\hbar^6+4\,x^2y^2\gamma\,\delta^2\hbar^6+4\,x\,y^2\gamma\,\delta\,\eta\,\hbar^6+4\,x^2y\,\gamma\,\delta\,\xi\,\hbar^6+4\,x\,y\,\gamma\,\eta\,\xi\,\hbar^6\big),\\
&\frac{1}{4\,(-\delta+T\delta+\hbar)^4}\,\epsilon\,\big(-8\,a\,T\,\delta^4\hbar+24\,a\,T^2\delta^4\hbar-24\,a\,T^3\delta^4\hbar+8\,a\,T^4\delta^4\hbar+2\,\gamma\,\delta^4\hbar-12\,T\,\gamma\,\delta^4\hbar+\\
&\quad 24\,T^2\gamma\,\delta^4\hbar-20\,T^3\gamma\,\delta^4\hbar+6\,T^4\gamma\,\delta^4\hbar+24\,a\,T\,\delta^3\hbar^2-48\,a\,T^2\delta^3\hbar^2+24\,a\,T^3\delta^3\hbar^2-4\,\gamma\,\delta^3\hbar^2+\\
&\quad 20\,T\,\gamma\,\delta^3\hbar^2-28\,T^2\gamma\,\delta^3\hbar^2+12\,T^3\gamma\,\delta^3\hbar^2+8\,a\,T\,x\,y\,\delta^4\hbar^2-16\,a\,T^2x\,y\,\delta^4\hbar^2+8\,a\,T^3x\,y\,\delta^4\hbar^2-\\
&\quad 8\,T\,x\,y\,\gamma\,\delta^4\hbar^2+16\,T^2x\,y\,\gamma\,\delta^4\hbar^2-8\,T^3x\,y\,\gamma\,\delta^4\hbar^2+8\,a\,T\,y\,\delta^3\eta\,\hbar^2-16\,a\,T^2y\,\delta^3\eta\,\hbar^2+\\
&\quad 8\,a\,T^3y\,\delta^3\eta\,\hbar^2+8\,a\,T\,x\,\delta^3\xi\,\hbar^2-16\,a\,T^2x\,\delta^3\xi\,\hbar^2+8\,a\,T^3x\,\delta^3\xi\,\hbar^2+8\,a\,T\,\delta^2\eta\,\xi\,\hbar^2-\\
&\quad 16\,a\,T^2\delta^2\eta\,\xi\,\hbar^2+8\,a\,T^3\delta^2\eta\,\xi\,\hbar^2-4\,\gamma\,\delta^2\eta\,\xi\,\hbar^2+20\,T\,\gamma\,\delta^2\eta\,\xi\,\hbar^2-28\,T^2\gamma\,\delta^2\eta\,\xi\,\hbar^2+\\
&\quad 12\,T^3\gamma\,\delta^2\eta\,\xi\,\hbar^2-24\,a\,T\,\delta^2\hbar^3+24\,a\,T^2\delta^2\hbar^3+2\,\gamma\,\delta^2\hbar^3-8\,T\,\gamma\,\delta^2\hbar^3+6\,T^2\gamma\,\delta^2\hbar^3-\\
&\quad 16\,a\,T\,x\,y\,\delta^3\hbar^3+16\,a\,T^2x\,y\,\delta^3\hbar^3+24\,T\,x\,y\,\gamma\,\delta^3\hbar^3-24\,T^2x\,y\,\gamma\,\delta^3\hbar^3+x^2y^2\gamma\,\delta^4\hbar^3+\\
&\quad 4\,T\,x^2y^2\gamma\,\delta^4\hbar^3-5\,T^2x^2y^2\gamma\,\delta^4\hbar^3-16\,a\,T\,y\,\delta^2\eta\,\hbar^3+16\,a\,T^2y\,\delta^2\eta\,\hbar^3-4\,y\,\gamma\,\delta^2\eta\,\hbar^3+\\
&\quad 16\,T\,y\,\gamma\,\delta^2\eta\,\hbar^3-12\,T^2y\,\gamma\,\delta^2\eta\,\hbar^3+8\,T\,x\,y^2\gamma\,\delta^3\eta\,\hbar^3-8\,T^2x\,y^2\gamma\,\delta^3\eta\,\hbar^3-y^2\gamma\,\delta^2\eta^2\hbar^3+\\
&\quad 4\,T\,y^2\gamma\,\delta^2\eta^2\hbar^3-3\,T^2y^2\gamma\,\delta^2\eta^2\hbar^3-16\,a\,T\,x\,\delta^2\xi\,\hbar^3+16\,a\,T^2x\,\delta^2\xi\,\hbar^3-4\,x\,\gamma\,\delta^2\xi\,\hbar^3+\\
&\quad 16\,T\,x\,\gamma\,\delta^2\xi\,\hbar^3-12\,T^2x\,\gamma\,\delta^2\xi\,\hbar^3+8\,T\,x^2y\,\gamma\,\delta^3\xi\,\hbar^3-8\,T^2x^2y\,\gamma\,\delta^3\xi\,\hbar^3-16\,a\,T\,\delta\,\eta\,\xi\,\hbar^3+\\
&\quad 16\,a\,T^2\delta\,\eta\,\xi\,\hbar^3+4\,\gamma\,\delta\,\eta\,\xi\,\hbar^3-16\,T\,\gamma\,\delta\,\eta\,\xi\,\hbar^3+12\,T^2\gamma\,\delta\,\eta\,\xi\,\hbar^3+8\,T\,x\,y\,\gamma\,\delta^2\eta\,\xi\,\hbar^3-\\
&\quad 8\,T^2x\,y\,\gamma\,\delta^2\eta\,\xi\,\hbar^3-x^2\gamma\,\delta^2\xi^2\hbar^3+4\,T\,x^2\gamma\,\delta^2\xi^2\hbar^3-3\,T^2x^2\gamma\,\delta^2\xi^2\hbar^3+\gamma\,\eta^2\xi^2\hbar^3-\\
&\quad 4\,T\,\gamma\,\eta^2\xi^2\hbar^3+3\,T^2\gamma\,\eta^2\xi^2\hbar^3+8\,a\,T\,\delta\,\hbar^4+8\,a\,T\,x\,y\,\delta^2\hbar^4-4\,x\,y\,\gamma\,\delta^2\hbar^4-12\,T\,x\,y\,\gamma\,\delta^2\hbar^4-\\
&\quad 4\,x^2y^2\gamma\,\delta^3\hbar^4-4\,T\,x^2y^2\gamma\,\delta^3\hbar^4+8\,a\,T\,y\,\delta\,\eta\,\hbar^4+4\,y\,\gamma\,\delta\,\eta\,\hbar^4-12\,T\,y\,\gamma\,\delta\,\eta\,\hbar^4-2\,x\,y^2\gamma\,\delta^2\eta\,\hbar^4-\\
&\quad 10\,T\,x\,y^2\gamma\,\delta^2\eta\,\hbar^4+2\,y^2\gamma\,\delta\,\eta^2\hbar^4-6\,T\,y^2\gamma\,\delta\,\eta^2\hbar^4+8\,a\,T\,x\,\delta\,\xi\,\hbar^4+4\,x\,\gamma\,\delta\,\xi\,\hbar^4-12\,T\,x\,\gamma\,\delta\,\xi\,\hbar^4-\\
&\quad 2\,x^2y\,\gamma\,\delta^2\xi\,\hbar^4-10\,T\,x^2y\,\gamma\,\delta^2\xi\,\hbar^4+8\,a\,T\,\eta\,\xi\,\hbar^4-16\,T\,x\,y\,\gamma\,\delta\,\eta\,\xi\,\hbar^4+2\,y\,\gamma\,\eta^2\xi\,\hbar^4-\\
&\quad 6\,T\,y\,\gamma\,\eta^2\xi\,\hbar^4+2\,x^2\gamma\,\delta\,\xi^2\hbar^4-6\,T\,x^2\gamma\,\delta\,\xi^2\hbar^4+2\,x\,\gamma\,\eta\,\xi^2\hbar^4-6\,T\,x\,\gamma\,\eta\,\xi^2\hbar^4+4\,x\,y\,\gamma\,\delta\,\hbar^5+\\
&\quad 4\,x^2y^2\gamma\,\delta^2\hbar^5+4\,x\,y^2\gamma\,\delta\,\eta\,\hbar^5+4\,x^2y\,\gamma\,\delta\,\xi\,\hbar^5+4\,x\,y\,\gamma\,\eta\,\xi\,\hbar^5\big)+\mathrm{Log}\Big[\frac{\hbar}{-\delta+T\delta+\hbar}\Big]\Big\}
\end{aligned}
$$

Reorderings with Rord

Rord

```
Rord_{u_i_, w_j_ → k_}[𝔈_U_[L___, {l___, u_i_, w_j_, r___}_s_, R___, Q_, P_]] :=
  Simp@Module[{υ, ω, δ, Λ1, yax, q, p, δ1 = ∂_{u_i, w_j} Q},
    {yax, q, p} = List @@ If[δ1 === 0, Λ_U[{υ, ω}, {u, w}], Λ_U[{υ, ω, δ}, {u, w}]] /.
      {y → y_k, a → a_k, x → x_k, t → t_s, T → T_s};
    𝔈_U[L, {l, Sequence @@ yax, r}_s, R, q + (Q /. u_i | w_j → 0), 𝕖^-q DP_{u_i→D_υ, w_j→D_ω}[P][p 𝕖^q]] /.
     {υ → ∂_{u_i} Q /. w_j → 0, ω → ∂_{w_j} Q /. u_i → 0, δ → δ1}]
```

```
With[{c𝕠 = 𝔈_CU[{y1, x1}_1, {x2, a2, y2}_2, ℏ t1 a2 + ℏ t1^-1 (𝕖^t1 - 1) y1 x2, 1 + ϵ x1 y2]},
 {Short[rhs = c𝕠 // Rord_{x2,a2→3}, 3], HL[CU[c𝕠] == CU[rhs]]}]
```

$\left\{\mathfrak{E}_{CU}\left[\{y_1, x_1\}_1, \{a_3, x_3, y_2\}_2, \frac{e^{-\gamma\hbar t_1}\left(e^{\gamma\hbar t_1}\hbar a_3 t_1^2 - \hbar x_3 y_1 + e^{t_1}\hbar x_3 y_1\right)}{t_1}, 1+\epsilon x_1 y_2\right], \text{True}\right\}$

```
With[{c𝕠 = 𝔈_CU[{y1, a1, x1}_1, {x2, a2, y2}_2,
    ℏ (l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2 + γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
    1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]},
 {Short[rhs = c𝕠 // Rord_{x2,a2→3}, 3], HL[CU[c𝕠] == CU[rhs]]}]
```

$\{\mathfrak{E}_{CU}[\{y_1, a_1, x_1\}_1, \{\ll 1\gg\}_2, \ll 1\gg \ll 1\gg,$
$1 + e^{-\gamma\hbar(l_{12}t_1+l_{22}t_2)}\,\epsilon\,(e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} a_1 l_1 + e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} a_3 l_2 + e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} p_{11} x_1 y_1 +$
$p_{21} x_3 y_1 + e^{\ll 1\gg} p_{12} x_1 y_2 + p_{22} x_3 y_2 - \gamma\hbar l_2 x_3 y_1 \gamma_{21} - \gamma\hbar l_2 x_3 y_2 \gamma_{22})], \text{True}\}$

```
With[{q𝕠 = 𝔈_QU[{y1, a1, x1}_1, {x2, a2, y2}_2,
    ℏ (l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2 + γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
    1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]},
 {Short[rhs = q𝕠 // Rord_{x2,a2→3}, 3], HL[QU[q𝕠] == QU[rhs]]}]
```

$\{\mathfrak{E}_{QU}[\{y_1, a_1, x_1\}_1, \{\ll 1\gg\}_2, \ll 1\gg \ll 1\gg,$
$1 + e^{-\gamma\hbar(l_{12}t_1+l_{22}t_2)}\,\epsilon\,(e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} a_1 l_1 + e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} a_3 l_2 + e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} p_{11} x_1 y_1 +$
$p_{21} x_3 y_1 + e^{\ll 1\gg} p_{12} x_1 y_2 + p_{22} x_3 y_2 - \gamma\hbar l_2 x_3 y_1 \gamma_{21} - \gamma\hbar l_2 x_3 y_2 \gamma_{22})], \text{True}\}$

```
With[{q𝕠 = 𝔈_QU[{y1, a1, x1}_1, {x2, a2, y2}_2,
    ℏ (l11 t1 a1 + l12 t1 a2 + l21 t2 a1 + l22 t2 a2 + γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
    1 + ϵ (l1 a1 + l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]},
 {Short[rhs = q𝕠 // Rord_{a2,y2→3}, 3], HL[QU[q𝕠] == QU[rhs]]}]
```

$\{\mathfrak{E}_{QU}[\{y_1, a_1, x_1\}_1, \{\ll 1\gg\}_2, \ll 1\gg \ll 1\gg,$
$1 + e^{-\gamma\hbar(l_{12}t_1+l_{22}t_2)}\,\epsilon\,(e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} a_1 l_1 + e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} a_3 l_2 + e^{\gamma\hbar(l_{12}t_1+l_{22}t_2)} p_{11} x_1 y_1 +$
$e^{\ll 1\gg} p_{21} x_2 y_1 + p_{12} x_1 y_3 + p_{22} x_2 y_3 - \gamma\hbar l_2 x_1 y_3 \gamma_{12} - \gamma\hbar l_2 x_2 y_3 \gamma_{22})], \text{True}\}$

```
With[{qΦ = 𝔈QU[{x1, y1}1, {x2, a2, y2}2,
    ℏ (l12 t1 a2 + l22 t2 a2 + γ11 x1 y1 + γ12 x1 y2 + γ21 x2 y1 + γ22 x2 y2),
    1 + ϵ (l2 a2 + p11 x1 y1 + p12 x1 y2 + p21 x2 y1 + p22 x2 y2)]},
 {Short[rhs = qΦ // Rord_{x1,y1→3}, 5], HL@SimpT[QU[qΦ] == QU[rhs]]}]
```

$\Big\{\mathbb{E}_{QU}\big[\{y_3, a_3, x_3\}_1, \{x_2, a_2, y_2\}_2, \frac{\hbar\, a_2\, l_{12}\, t_1 + \ll 16 \gg + \hbar\, T_1\, x_2\, y_2\, \gamma_{11}\, \gamma_{22}}{1 - \gamma_{11} + T_1\, \gamma_{11}},$

$\frac{1}{1-\gamma_{11}+T_1\gamma_{11}} + \Big(\epsilon\,\big(4\,\hbar\, a_2\, l_2 + 4\, p_{11} - 4\, p_{11}\, T_1 + 4\,\hbar\, p_{22}\, x_2\, y_2 + \ll 339 \gg + \gamma\,\hbar^4\, x_2^2\, y_2^2\, \gamma_{12}^2\, \gamma_{21}^2 -$

$4\,\gamma\,\hbar^4\, T_1\, x_2^2\, y_2^2\, \gamma_{12}^2\, \gamma_{21}^2 + 3\,\gamma\,\hbar^4\, T_1^2\, x_2^2\, y_2^2\, \gamma_{12}^2\, \gamma_{21}^2\big)\Big) \Big/ \Big(4\,\hbar\,\big(1 - \gamma_{11} + T_1\,\gamma_{11}\big)^5\Big)\big], \text{True}\Big\}$

R in QU.

Faddeev-Quesne's formula:

Faddeev

$$\mathbb{e}_{q\_,k\_}[x\_] := \mathbb{e}^\wedge\left(\sum_{j=1}^{k} \frac{(1-q)^j\, x^j}{j\,(1-q^j)}\right);\ \ \mathbb{e}_{q\_}[x\_] := \mathbb{e}_{q,\$k}[x]$$

```
Table[Together@SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}], {n, 0, 5}]
```

$\left\{1, 1, \frac{1}{1+\rho}, \frac{1}{(1+\rho)\,(1+\rho+\rho^2)}, \frac{1}{(1+\rho)^2\,(1+\rho^2)\,(1+\rho+\rho^2)},\right.$
$\left. 1\Big/\left((1+\rho)^2\,(1+\rho^2)\,(1+\rho+\rho^2)\,(1+\rho+\rho^2+\rho^3+\rho^4)\right)\right\}$

```
Table[HL@FunctionExpand[QFactorial[n, ρ] SeriesCoefficient[𝕖_{ρ,5}[x], {x, 0, n}]], {n, 0, 5}]
```

{1, 1, 1, 1, 1, 1}

R

```
QU[R_{i_,j_}] := 𝕆QU[{y1, a1}i, {a2, x2}j, SS[𝕖^{ℏ b1 a2} 𝕖q[ℏ y1 x2] /. b1 → γ^-1 (ϵ a1 - ti)]];
QU[R^-1_{i_,j_}] := Sj@QU[R_{i,j}];
```

```
QU[R_{3,4}] // Short
```

$QU[\,] + \frac{\epsilon\,\hbar\, QU[a_3, a_4]}{\gamma} + \hbar\, QU[y_3, x_4] + \frac{\ll 1 \gg}{\gamma} +$
$\ll 1 \gg - \frac{\ll 1 \gg}{\gamma} - \frac{\epsilon \ll 3 \gg}{\gamma^2} - \frac{\hbar^2\, QU[y_3, a_4, x_4]\, t_3}{\gamma} + \frac{\hbar^2\, QU[a_4, a_4]\, t_3^2}{2\,\gamma^2}$

Verifying R2 (~2 secs @ $p=4, $k=2):

```
QU[R_{1,2} ** R^-1_{1,2}] // Simp // HL // Timing
```

{0.0625, QU[]}

Verifying R3 (~156 secs @ $p=4, $k=2):

```
{Short[lhs = QU[R1,2 ** R1,3 ** R2,3]], HL@SimpT[lhs - QU[R2,3 ** R1,3 ** R1,2]]} // Timing
```

$\left\{0.34375, \left\{QU[\,] + \frac{\epsilon\,\hbar\,QU[a_1, a_2]}{\gamma} + \ll 85 \gg + QU[y_1, y_1, x_3, x_3]\left(\frac{\hbar^2}{2} - \hbar^2\,T_2 + \frac{1}{2}\hbar^2\,T_2^2\right), 0\right\}\right\}$

R in $Œ_{QU}$.

RinOE

```
ŒQU[Ri_,j_] := ŒQU[{yi, ai, xi}i, {yj, aj, xj}j, -ħ γ^-1 ti aj + ħ yi xj,
  Normal@Series[e^(ħ γ^-1 ti aj - ħ yi xj) (e^(ħ bi aj) eq[ħ yi xj] /. bi → γ^-1 (ϵ ai - ti)), {ϵ, 0, $k}]]
```

```
ŒQU[R1,2]
```

$Œ_{QU}\Big[\{y_1, a_1, x_1\}_1, \{y_2, a_2, x_2\}_2, -\frac{\hbar\,a_2\,t_1}{\gamma} + \hbar\,x_2\,y_1,$
$1 + \epsilon\left(\frac{\hbar\,a_1\,a_2}{\gamma} - \frac{1}{4}\gamma\,\hbar^3\,x_2^2\,y_1^2\right) + \frac{1}{2}\epsilon^2\left(\frac{\hbar\,a_1\,a_2}{\gamma} - \frac{1}{4}\gamma\,\hbar^3\,x_2^2\,y_1^2\right)^2\Big]$

𝔼

𝔼[L, Q, P] means $e^{\hbar(L+Q)}P$, where L is linear in the a's, Q is a combination of $x_i\,y_j$, and P is a perturbation polynomial. It should be interpreted via C𝕆[𝔼[...], $\{x_1,\ a_1,\ y_1\}_i$, ...] (with some default for direct interpretation), or likewise via Q𝕆[𝔼[...], $\{x_1,\ a_1,\ y_1\}_i$, ...]. In themselves, C𝕆 and Q𝕆 should have an interpretation in CU/QU by casting.

Alternative Algorithms

```
λalt[CU] := Module[{eq, d, b, c, so},
   eq = ρ@e^(ξ xCU).ρ@e^(η yCU) == ρ@e^(d yCU).ρ@e^(c (t 1CU - 2 ϵ aCU)).ρ@e^(b xCU);
   {so} = Solve[Thread[Flatten /@ eq], {d, b, c}] /. C@1 → 0;
   Normal@Series[e^(-η y - ξ x + η ξ t + c t + d y - 2 ϵ c a + b x) /. so, {ϵ, 0, $k}]];
```

```
{λalt[CU], HL@Simplify[λalt[CU] == Last[ΛCU[{ξ, η}, {x, y}]]]}
```

$\left\{1 + \epsilon\left(2\,a\,\eta\,\xi - y\,\gamma\,\eta^2\,\xi - x\,\gamma\,\eta\,\xi^2 + \frac{1}{2}t\,\gamma\,\eta^2\,\xi^2\right), \text{True}\right\}$